
       Identification Division.
       Program-ID.    J7200524.
*     * 
*     * Copyright Wisconsin Department of Transportation
*     * 
*     * Permission is hereby granted, free of charge, to any person or
*     * organisation to use this software and its associated files
*     * subject to the following conditions:
*     * 
*     * The software may be redistributed free of charge to any other
*     * person or organisation provided that the above copyright 
*     * notice, this permission notice and the disclaimer shall be 
*     * included with all copies of the Software.
*     * 
*     * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF
*     * ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED
*     * TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A
*     * PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT
*     * SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR
*     * ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN
*     * ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
*     * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
*     * OTHER DEALINGS IN THE SOFTWARE. IF THIS DISCLAIMER
*     * CONTRADICTS LOCALLY APPLICABLE LAW THEN USE OF THIS
*     * SOFTWARE IS PROHIBITED.
*     * 
      *
      * This program wraps up the LE callable services necessary to
      * obtain the current date and time (GMT or local) and format
      * that date and/or time according to a format string.  This
      * is conceptually similar to the C function strftime().
      *
      * Valid values for the format string can be found by looking
      * up the CEEDATM callable service in the Language Environment
      * Programming Reference.
      *
      * If errors are encountered in any of the LE callable services,
      * the service will display a message.  It appears that CEEMSG
      * is used for this purpose.
      *
      * Under systems other than CICS, Language Environment writes
      * the message to the ddname of the file specified in the MSGFILE
      * run-time option. Under CICS, the message is written to a
      * transient data queue named CESE.
      *
      * Currently, this means that in batch the message is likely to



      * be found in SYSOUT and in CICS it's likely to be found in
      * CEEMSG.
      *
      *Working-Storage Section.
      *01  CONSTANTS.
      *    05  MYNAME                  PIC X(008) VALUE 'J8675309'.
      *    05  FMT-DT-TM-PGM           PIC X(008) VALUE 'J7200524'.
      *
      *01  LCL-TM-SW               PIC X(001).
      *    88  LCL-TM                         VALUE 'Y'.
      *    88  GMT-TM                         VALUE 'N'.
      *
      *01  DT-PIC-STRN.
      *    05  DT-PIC-STRN-LEN     PIC S9(004) COMP-5 VALUE +80.
      *    05  DT-PIC-STRN-VAL     PIC X(080)
      *        VALUE 'Www, DD Mmm YYYY HH:MI:SS GMT'.
      *
      *01  FMT-DT-TM               PIC X(080) VALUE SPACES.
      *
      *Procedure Division.
      *    SET GMT-TM TO TRUE
      *    CALL J7200524-PGM USING
      *        LCL-TM-SW
      *        DT-PIC-STRN
      *        FMT-DT-TM
      *    END-CALL
      *

       Environment Division.
       Data Division.
       Working-Storage Section.
       01  CONSTANTS.
           05  MYNAME                  PIC X(008) VALUE 'J7200524'.

       01  WORK-AREAS.
           05  GMT-LILIAN              PIC 9(008)  COMP-5 VALUE 0.
           05  OSET-HRS                PIC S9(008) COMP-5 VALUE +0.
           05  OSET-MIN                PIC S9(008) COMP-5 VALUE +0.
           05  GMT-FLOT                            COMP-2 VALUE 0.0.
           05  OSET-SEC                            COMP-2 VALUE 0.0.

       Linkage Section.
      * Input -  Switch indicating whether to pass back local time
      *          or GMT.  A value of 'Y' in this field indicates
      *          local time should be used.  Any value other than 'Y'
      *          indicates GMT should be used.



       01  LCL-TM-SW               PIC X(001).
           88  LCL-TM                         VALUE 'Y'.

      * Input -  Date picture string prefixed by its length.  Valid
      *          picture string formatting is documented in the
      *          Language Environment Programming Reference.
       01  DT-PIC-STRN.
           05  DT-PIC-STRN-LEN     PIC S9(004) COMP-5.
           05  DT-PIC-STRN-VAL
               OCCURS 1 TO 80
               DEPENDING DT-PIC-STRN-LEN
                                   PIC X(001).

      * Output - GMT date/time formatted according to the content
      *          of DT-PIC-STRN-VAL.
       01  FMT-DT-TM               PIC X(080).

       Procedure Division Using
           LCL-TM-SW
           DT-PIC-STRN
           FMT-DT-TM
           .

           MOVE SPACES TO FMT-DT-TM

      *    Obtain GMT time.
           CALL 'CEEGMT' USING
               GMT-LILIAN
               GMT-FLOT
               OMITTED
           END-CALL

           IF LCL-TM
      *        If local time requested, obtain its offset from GMT
      *        and compute local time.
               CALL 'CEEGMTO' USING
                   OSET-HRS
                   OSET-MIN
                   OSET-SEC
                   OMITTED
               END-CALL
               ADD OSET-SEC TO GMT-FLOT
           END-IF

      *    Format the date according to the passed picture string.
           CALL 'CEEDATM' USING



               GMT-FLOT
               DT-PIC-STRN
               FMT-DT-TM
               OMITTED
           END-CALL

           MOVE 0       TO RETURN-CODE
           GOBACK.


